

# N=7\_Algebraic\_proof

July 20, 2021

---

## MULTIPLE EXTREMAL DISC-PACKINGS IN COMPACT HYPERBOLIC SURFACES

### \*\* CASE N = 7: ALGEBRAIC PROOF\*\*

**An NEC-group uniformizing a non-orientable surface of genus p=3 that contains at least two extremal 6-packings** Summary of the particular data for this file. It all comes from the numerical experiments:

- N=7 in 1)
  - A particular set of side-pairing transformations in 6)
  - The two bananas: side-pairing 5, banana 1, sign -1 and side-pairing 8, banana 2, sign +1 in 8)
  - The point z~0.0353+0.4034i is the second intersection point in 9) (indexed 1, since sagemath first index is 0)
  - The point w in 11) is not the center of the disc. The exponent j equals 3.
  - A particular set of products of side-pairings in 14)
- 

Recall that  $u = e^R$  where  $\cosh R = \frac{1}{2 \sin \frac{\pi}{N}}$ .

Therefore  $u + u^{-1} = \frac{u^2 + 1}{u} = \frac{1}{\sin \frac{\pi}{N}}$ , thus  $\sin \frac{\pi}{N} u^2 - u + \sin \frac{\pi}{N} = 0$  and we have

$$u = \frac{1 + \sqrt{1 - 4 \sin^2 \frac{\pi}{N}}}{2 \sin \frac{\pi}{N}}.$$

```
[23]: #Approximate value of u
N=7
print (e^(arccosh(1/(2*sin(pi/N))))).n(), (( 1+sqrt( 1-4*(sin(pi/N))^2 ) )/
→(2*sin(pi/N))).n()
```

1.72508242466029 1.72508242466029

---

1) We compute the minimal polynomial of  $u$ , assigned to the variable  $\text{pol}$ .

[24] :

```
N=7  
pol=(( 1+sqrt( 1-4*(sin(pi/N))^2 ) )/(2*sin(pi/N))).minpoly()  
show(pol)
```

```
x^12 - 2*x^10 - x^8 - 36/7*x^6 - x^4 - 2*x^2 + 1
```

[25] : #Approximate value of the roots of  $\text{pol}$ : notice that  $u$  is the largest real root  
`show(pol.roots(ring=QQbar))`

```
[(-1.725082424660292?, 1),  
 (-0.5796824463021952?, 1),  
 (0.5796824463021952?, 1),  
 (1.725082424660292?, 1),  
 (-0.6395240038449663? - 0.7687711288193018?*I, 1),  
 (-0.6395240038449663? + 0.7687711288193018?*I, 1),  
 (-0.5128584316362769? - 0.8584731964945546?*I, 1),  
 (-0.5128584316362769? + 0.8584731964945546?*I, 1),  
 (0.5128584316362769? - 0.8584731964945546?*I, 1),  
 (0.5128584316362769? + 0.8584731964945546?*I, 1),  
 (0.6395240038449663? - 0.7687711288193018?*I, 1),  
 (0.6395240038449663? + 0.7687711288193018?*I, 1)]
```

---

2) We denote the field  $\mathbb{Q}(u)$  as  $K1$ .

[26] : `K1.<u> = NumberField(pol,embedding=2)  
CC(u)`

#The instruction  $\text{CC}$  gives the approximate value of an element of the field  $K1$ .  
#The given embedding into the reals is chosen so that  $\text{CC}(u)$  gives the right value

[26] : 1.72508242466029

---

3) Since  $u = e^R$ , we see that  $\sin \frac{\pi}{N}$  is the field element given by  $\frac{u}{u^2 + 1}$ . We call it  $\text{sn}$ .

[27] : `print(CC(u/(u^2+1)),sin(pi/N).n()) #A numerical verification  
sn=u/(u^2+1)  
show(sn) #Displays sn as an element of K1`

```
(0.433883739117551, 0.433883739117558)
```

```
-7/64*u^11 + 21/64*u^9 - 7/32*u^7 + 25/32*u^5 - 43/64*u^3 + 57/64*u
```

4) Accordingly,  $\cos^2 \frac{\pi}{N} = 1 - \sin^2 \frac{\pi}{N} = 1 - \frac{u^2}{(u^2+1)^2}$  is the field element given by  $\frac{u^4+u^2+1}{(u^2+1)^2}$ .

We call  $v = \cos \frac{\pi}{N}$ . It is an element of  $\mathbb{Q}(u)$  if and only if  $u^4 + u^2 + 1$  is a square in K1. We also denote it as cs as a nickname.

We denote  $L = \mathbb{Q}(u, v)$ . This field agrees with K1 when  $v \in K1$ , of course.

[28]: `print(CC(sqrt(u^4+u^2+1)/(u^2+1)),(cos(pi/N)).n()) #Numerical verification`

```

if (u^4+u^2+1).is_square():
    v=sqrt(u^4+u^2+1)/(u^2+1) # cos(pi/N)
    print(CC(v))
    cs=v
    show(v) #Displays v as an element of K1
    L=K1
else:
    Ring.<t> = PolynomialRing(K1)
    L.<v> = K1.extension(t^2-(u^4+u^2+1)/((u^2+1)^2))
    cs=v

#The code below this line constructs the right embedding of L into the complex
#numbers
#that matches the numerical values we know that u and v must have.
#We call this embedding L_approx
#It will be very important later, in particular in points 8) and 11)
j=0
test=0
while test!=1:
    L_approx = L.embeddings(CC)[j]
    if abs(CC(sqrt(u^4+u^2+1)/(u^2+1))-L_approx(v))<10^-3 and
    abs(CC(u)-L_approx(u))<10^(-3): test+=1
    j+=1
print L_approx(u),L_approx(v) #This line must give the values we knew in advance
#(numerical verification)

```

(0.900968867902414, 0.900968867902419)

0.900968867902414

$-7/16*u^8 + 7/8*u^6 + 7/8*u^4 + 11/8*u^2 + 5/16$

1.72508242466029 0.900968867902414

5) The matrices of  $a, b, c$  are

$$a = \begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}, \quad b = \begin{pmatrix} 0 & u^{-1} \\ u & 0 \end{pmatrix}, \quad c = \begin{pmatrix} -\cos \frac{\pi}{N} & \sin \frac{\pi}{N} \\ \sin \frac{\pi}{N} & \cos \frac{\pi}{N} \end{pmatrix}$$

```
[29]: a=matrix(2,[1,0,0,-1])  
b=matrix(2,[0,u^(-1),u,0])  
c=matrix(2,[-cs,sn,sn,cs])  
show(a)  
show(b)  
show(c)
```

6) We introduce in the list  $s$  the side-pairings for this surface group: our conjecture, driven by numerical experiments, is that the surface group thus generated will contain more than one extremal packing.

We use the notation  $G_i = (ca)^i(abcb)(ca)^{-i}$ .

```
[30]: def G(i):
    return (c*a)^i*(a*b*c*b)*(c*a)^(-i)

s=['zero'] #for s[j] to denote the j-th side-pairing we add an "empty" item in
           #position 0
s.append(G(1)^4*(c*a*b)*((c*a)^(-1))*G(1)^4) #s1
s.append(G(1)^5*(c*a*a*b)*((c*a)^(-1))*G(1)^5) #s2
s.append(G(2)^2*(c*a*c*a*a*b)*((c*a)^(-1))*G(1)^(-1)) #s3
s.append(G(2)^3*(c*a*c*a)) #s4
s.append(G(3)^(-2)*(c*a*c*a*c*a*b)*((c*a)^(-2))*G(2)^3) #s5
s.append(G(4)^2*(c*a*c*a*c*a*c*a*b)*((c*a)^(-2))*G(2)^2) #s6
s.append(G(4)^3*(c*a*c*a*c*a*c*a*b)*((c*a)^(-3))*G(3)^(-2)) #s7
s.append(G(5)^(-1)*(c*a*c*a*c*a*c*a*c*a*b)*((c*a)^(-3))*G(3)^(-3)) #s8
s.append(((c*a)^6)*(a)*((c*a)^(-3))*G(3)^3) #s9
s.append(((G(5)^4)*(G(4)^2))) #s10
s.append(G(5)^2*(G(4)^(-1))*(c*a*c*a*c*a*c*a*a)*((c*a)^(-4))*G(4)^2) #s11
s.append(G(5)^4*(c*a*c*a*c*a*c*a*c*a*b)*((c*a)^(-5))*G(5)^(-2)) #s12
```

```
show(s[1:]) # How do the matrices look like? Their coefficients lie in L, but
→are they too horrible?
```

```
[  
[ -21/64*u^10 + 21/64*u^8 + 35/32*u^6 + 61/32*u^4 + 143/64*u^2 + 49/64 -47/  
-32*u^11 + 73/32*u^9 + 41/16*u^7 + 993/112*u^5 + 141/32*u^3 + 137/32*u] [ ]  
-21/64*u^10 + 63/64*u^8 - 7/32*u^6 + 33/32*u^4 - 45/64*u^2 + 15/64 45/64*u^11 -  
83/64*u^9 - 19/32*u^7 - 957/224*u^5 - 151/64*u^3 - 135/64*u] [ ] -21/  
-64*u^10 + 63/64*u^8 - 7/32*u^6 + 33/32*u^4 - 45/64*u^2 + 15/64 45/64*u^11 - 83/  
-64*u^9 - 19/32*u^7 - 957/224*u^5 - 151/64*u^3 - 135/64*u] [ ] 21/64*u^10  
-63/64*u^8 + 7/32*u^6 - 33/32*u^4 + 45/64*u^2 - 15/64 -45/64*u^11 + 83/64*u^9  
+ 19/32*u^7 + 957/224*u^5 + 151/64*u^3 + 135/64*u] [ ] 7/64*u^10 - 35/  
-64*u^8 + 21/32*u^6 + 3/32*u^4 + 31/64*u^2 - 19/64 37/64*u^11 - 95/64*u^9 - 1/  
-32*u^7 - 519/224*u^5 + 29/64*u^3 - 31/64*u] [ ] 7/64*u^10 - 35/64*u^8 +  
-21/32*u^6 + 3/32*u^4 + 31/64*u^2 - 19/64 37/64*u^11 - 95/64*u^9 - 1/32*u^7 -  
-519/224*u^5 + 29/64*u^3 - 31/64*u] [ ] 7/64*u^10 - 35/64*u^8 + 21/32*u^6  
+ 3/32*u^4 + 31/64*u^2 - 19/64 37/64*u^11 - 95/64*u^9 - 1/32*u^7 - 519/  
-224*u^5 + 29/64*u^3 - 31/64*u] [ ] 7/32*u^10 - 7/8*u^6 - 39/  
-16*u^4 - 51/32*u^2 - 29/16 31/64*u^11 - 69/64*u^9 - 19/32*u^7 - 313/224*u^5 -  
-9/64*u^3 - 45/64*u] [ ] -7/32*u^10 + 7/8*u^6 + 39/16*u^4 + 51/  
-32*u^2 + 29/16 -31/64*u^11 + 69/64*u^9 + 19/32*u^7 + 313/224*u^5 + 9/64*u^3 +  
-45/64*u] [21/32*u^10 - 21/32*u^8 - 35/16*u^6 - 61/16*u^4 - 87/32*u^2 - 57/32  
-7/16*u^11 + 21/32*u^9 + 7/8*u^7 + 43/16*u^5 + 25/16*u^3 + 21/32*u] [ ]  
-7/64*u^10 - 7/64*u^8 + 7/32*u^6 - 67/32*u^4 - 85/64*u^2 - 115/64 7/  
-64*u^11 - 7/64*u^9 - 35/32*u^7 + 31/32*u^5 - 1/64*u^3 + 105/64*u] [ ] -7/  
-64*u^10 + 7/64*u^8 - 7/32*u^6 + 67/32*u^4 + 85/64*u^2 + 115/64  
-64*u^11 + 7/64*u^9 + 35/32*u^7 - 31/32*u^5 + 1/64*u^3 - 105/64*u]
```

```

[      21/32*u^11 - 35/32*u^9 - 21/16*u^7 - 47/16*u^5 - 43/32*u^3 - 15/32*u      □
 $\downarrow$  35/64*u^10 - 63/64*u^8 - 21/32*u^6 - 111/32*u^4 - 57/64*u^2 - 99/64], [      7/
 $\downarrow$  64*u^11 - 49/64*u^9 + 35/32*u^7 + 3/32*u^5 + 131/64*u^3 - 5/64*u      -35/
 $\downarrow$  64*u^10 + 77/64*u^8 + 7/32*u^6 + 83/32*u^4 + 69/64*u^2 + 93/64], [      7/
 $\downarrow$  64*u^11 - 49/64*u^9 + 35/32*u^7 + 3/32*u^5 + 131/64*u^3 - 5/64*u      -35/
 $\downarrow$  64*u^10 + 77/64*u^8 + 7/32*u^6 + 83/32*u^4 + 69/64*u^2 + 93/64], [      -7/
 $\downarrow$  64*u^11 + 49/64*u^9 - 35/32*u^7 - 3/32*u^5 - 131/64*u^3 + 5/64*u      35/
 $\downarrow$  64*u^10 - 77/64*u^8 - 7/32*u^6 - 83/32*u^4 - 69/64*u^2 - 93/64], [77/64*u^11 - □
 $\downarrow$  119/64*u^9 - 77/32*u^7 - 205/32*u^5 - 243/64*u^3 - 239/64*u      7/64*u^10 + □
 $\downarrow$  21/64*u^8 - 35/32*u^6 - 53/32*u^4 - 145/64*u^2 - 59/64], [77/64*u^11 - 119/
 $\downarrow$  64*u^9 - 77/32*u^7 - 205/32*u^5 - 243/64*u^3 - 239/64*u      7/64*u^10 + 21/
 $\downarrow$  64*u^8 - 35/32*u^6 - 53/32*u^4 - 145/64*u^2 - 59/64], [77/64*u^11 - 119/64*u^9 □
 $\downarrow$  - 77/32*u^7 - 205/32*u^5 - 243/64*u^3 - 239/64*u      7/64*u^10 + 21/64*u^8 - □
 $\downarrow$  35/32*u^6 - 53/32*u^4 - 145/64*u^2 - 59/64], [63/64*u^11 - 133/64*u^9 - 35/
 $\downarrow$  32*u^7 - 127/32*u^5 - 41/64*u^3 - 77/64*u      7/32*u^10 - 7/8*u^8 + 7/
 $\downarrow$  8*u^6 - 11/16*u^4 + 37/32*u^2 - 3/16], [-63/64*u^11 + 133/64*u^9 + 35/32*u^7 + □
 $\downarrow$  127/32*u^5 + 41/64*u^3 + 77/64*u      -7/32*u^10 + 7/8*u^8 - 7/8*u^6 + 11/
 $\downarrow$  16*u^4 - 37/32*u^2 + 3/16], [ 21/32*u^11 - 7/8*u^9 - 21/16*u^7 - 17/4*u^5 - 107/
 $\downarrow$  32*u^3 - 23/8*u      -21/32*u^10 + 49/32*u^8 + 7/16*u^6 + 33/16*u^4 - 1/32*u^2 + 5/
 $\downarrow$  32], [105/64*u^11 - 217/64*u^9 - 49/32*u^7 - 235/32*u^5 - 167/64*u^3 - 209/64*u □
 $\downarrow$  -49/64*u^10 + 77/64*u^8 + 63/32*u^6 + 105/32*u^4 + 115/64*u^2 + 33/64], □
 $\downarrow$  [-105/64*u^11 + 217/64*u^9 + 49/32*u^7 + 235/32*u^5 + 167/64*u^3 + 209/64*u      □
 $\downarrow$  49/64*u^10 - 77/64*u^8 - 63/32*u^6 - 105/32*u^4 - 115/64*u^2 - 33/64]
]
```

---

7) We compute  $\cosh^2 \frac{d_n}{2}$  for the distances  $d_0, d_1, d_2$  associated to a tiling of hyperbolic space with regular N-gons of angle  $2\pi/3$ . We store the three values in the list ch.

First, as  $d_0 = 2R$ , then  $\cosh^2 \frac{d_0}{2} = \cosh R = \frac{(u^2+1)^2}{4u^2}$ .

Second,  $\cosh d_1 = \cosh^2(d_0) - \sinh^2(d_0) \cos(4\pi/N) = (1 - \cos(4\pi/N)) \cosh^2(d_0) + \cos(4\pi/N)$ .

Now, since  $\cosh d = 2 \cosh^2 \frac{d}{2} - 1$ , we have

$$\cosh^2 \left( \frac{d_1}{2} \right) = \frac{1 + \cosh d_1}{2} = \frac{1 + (1 - \cos(4\pi/N)) \cosh^2(d_0) + \cos(4\pi/N)}{2}$$

that is

$$\cosh^2 \left( \frac{d_1}{2} \right) = \frac{1 + \cosh d_1}{2} = \frac{1 + (1 - \cos(4\pi/N))(2 \cosh^2 \frac{d_0}{2} - 1)^2 + \cos(4\pi/N)}{2}.$$

Accordingly, we have  $\cosh d_2 = \cosh^2(d_0) - \sinh^2(d_0) \cos(6\pi/N) = (1 - \cos(6\pi/N)) \cosh^2(d_0) + \cos(6\pi/N)$  and therefore

$$\cosh^2\left(\frac{d_2}{2}\right) = \frac{1 + \cosh d_2}{2} = \frac{1 + (1 - \cos(6\pi/N))(2 \cosh^2 \frac{d_0}{2} - 1)^2 + \cos(6\pi/N)}{2}.$$

**NOTE:** Recall that  $\cos \frac{k\pi}{N}$  is the value at  $x = \cos \frac{\pi}{N}$  of the  $k$ -th Chebyshev polynomial.

[31]:

```
ch=[(u^2+1)^2/(4*u^2)]
ch.append((1+(1-chebyshev_T(4,cs))*(2*ch[0]-1)^2+chebyshev_T(4,cs))/(2))
ch.append((1+(1-chebyshev_T(6,cs))*(2*ch[0]-1)^2+chebyshev_T(6,cs))/(2))

show(ch) # How do these values look like? They lie in L, but are they too
         →horrible?
```

---


$$[-1/4u^{10} + 1/2u^8 + 1/4u^6 + 9/7u^4 + 1/2u^2 + 1, \\ -1/8u^{10} - 3/16u^8 + u^6 + 85/56u^4 + 13/8u^2 + 21/16, \\ -1/2u^{10} + u^8 + 1/2u^6 + 18/7u^4 + u^2 + 2]$$

8) We compute the equations of the two relevant bananas, and the coordinates of the intersection. We need to recall which bananas are the promising ones in each case (meaning for which side-pairing and which distance  $d_j$ ) from the information we obtained numerically.

The two variables  $sign$  can be (independently) 1 or  $-1$ : we guess the right choice after computing (via CC) the numerical value of the intersection, that we expect to match what we obtained in our numerical experiments.

[32]:

```
def H(k,s): #A function that occurs in the equation of the banana
    if s.det() == -1:
        return 4*ch[k]
    elif s.det() == 1:
        return 4*(ch[k] - 1)

def banana(sign, s, k): #The equation itself: choosing sign=+1 or -1 gives each
    →of the two arcs
    polring=PolynomialRing(L, 'x,y')
    x, y=polring.0, polring.1
    A,B,C,D=s[0][0], s[0][1], s[1][0], s[1][1]
    return
    →sign*((B-C)*(1+x^2+y^2)+2*x*(A-D)-2*y*(B+C))-(1-x^2-y^2)*sqrt(H(k,s)-(A+D)^2+4*det(s))

##### Defining the equations
var('x,y')
sign1,sign2=-1,1 #Choose the right couple of signs to point to the right
         →component of each banana!
```

```

eq1=banana(sign1, s[5], 1) #Insert here the right number of sidepairing s[j] for this case!!!!!
eq2=banana(sign2, s[8], 2) #The third number is either 0, 1, or 2 and indicates which banana we want.
                                         #Choose the right ones for this case!!!
show(eq1)
show(eq2)
#####

##### A bit of algebraic geometry here: J is the variety determined by both polynomials. We check first that
##### it has dimension 0: it consists of finitely many points. Finally we compute those points.
polring=PolynomialRing(L, 'x,y')
J = polring.ideal(eq1, eq2)
print J.dimension()
J.variety(L)

```

```

(7/16*u^11 - 7/16*u^9 - 7/8*u^7 - 25/8*u^5 - 57/16*u^3 - 71/16*u)*x^2 + (7/16*u^11 - 7/16*u^9 - 7/8*u^7 - 25/8*u^5 - 57/16*u^3 - 71/16*u)*y^2 + (7/4*u^8 - 7/2*u^6 - 7/2*u^4 - 11/2*u^2 - 5/4)*x + (57/16*u^11 - 107/16*u^9 - 39/8*u^7 - 977/56*u^5 - 107/16*u^3 - 135/16*u)*y + (13/16*u^11 - 5/16*u^9 - 31/8*u^7 - 283/56*u^5 - 79/16*u^3 - 33/16*u)

(7/4*u^8 - 7/2*u^6 - 7/2*u^4 - 11/2*u^2 - 13/4)*x + (-47/16*u^11 + 101/16*u^9 + 27/8*u^7 + 601/56*u^5 + 25/16*u^3 + 61/16*u)*y + (-u^11 + 2*u^9 + u^7 + 36/7*u^5 + u^3 + u)

0

```

[32]: [ {y: 21/32\*u^10 - 35/32\*u^8 - 21/16\*u^6 - 47/16\*u^4 - 43/32\*u^2 - 31/32, x: 7/64\*u^11 - 7/64\*u^9 - 7/32\*u^7 - 25/32\*u^5 - 57/64\*u^3 - 71/64\*u}, {y: 35/128\*u^10 - 63/128\*u^8 - 35/64\*u^6 - 69/64\*u^4 - 29/128\*u^2 - 23/128, x: -19/128\*u^11 + 59/128\*u^9 - 15/64\*u^7 + 293/448\*u^5 - 47/128\*u^3 - 1/128\*u} ]

---

9) We apply L\_approx to the points we have obtained in order to determine which is the one that corresponds to the value we expected from our numerical experiments. If none of them seems to be correct we need to go back to 8) and change the signs.

[33]: #WARNING: The numerical experiments gave z~0.0353+0.4034i in this surface  
for point in J.variety(L):  
 print L\_approx(point[x]), L\_approx(point[y])

---

```

0.873057463676130 1.81292250376066
0.0353613040941071 0.403413208727246

```

---

10) We call  $zx$  and  $zy$  the real and imaginary part of the right intersection of bananas. The point  $z = zx + izy$  is, still conjecturally, the center of a disc of a second (hidden) extremal packing.

[34] : `zx, zy=J.variety(L)[1][x], J.variety(L)[1][y]`  
*#Choose J.variety(L)[S] for the right value of S checking the output of 9)*

```
show(zx) #Again having a look. Are they very large expressions in u?
show(zy)
```

---

```
-19/128*u^11 + 59/128*u^9 - 15/64*u^7 + 293/448*u^5 - 47/128*u^3 - 1/128*u
35/128*u^10 - 63/128*u^8 - 35/64*u^6 - 69/64*u^4 - 29/128*u^2 - 23/128
```

---

11) Our numerical experiments suggest that the following center of the initial k-packing will have an important role:

$$w = \frac{u^2 - 1}{u^2 + 1} \left( \sin \frac{2\pi j}{N} - i \cos \frac{2\pi j}{N} \right)$$

#### for certain integer  $j$  (we need to get the right one in each case from the experiments).

(there is also an special case possible in which  $w = 0$ , the center of the unit disc).

We compute  $\cos \frac{2\pi j}{N}$  as the value at  $x = \cos \frac{\pi}{N}$  of the  $2j$ -th Chebyshev polynomial. Then, we compute  $\sin \frac{2\pi j}{N} = \sqrt{1 - \cos^2 \frac{2\pi j}{N}}$ .

Important.- By the last expression we mean *the positive root*, but we don't know which of the two possible roots will Sage associate to the symbols  $\sqrt{\phantom{x}}$  and  $-\sqrt{\phantom{x}}$  applied to a given element of the field L. We need again to make use of L\_approx to decide which one is the positive one (the warning is motivated for the fact that it may happen that the positive one is the one sage calls  $-\sqrt{\phantom{x}}$ ).

[35] : `#w_is_center='yes' #leave uncommented the right line. w_is_center='yes' si w is
→the center of the unit disc
w_is_center='no'`

```
if w_is_center=='yes':
    wx=0
    wy=0
elif w_is_center=='no':
    j=3 #Enter here the right value for j in this case
    wy=-(u^2-1)*chebyshev_T(2*j,cs)/(u^2+1)
    wx=(u^2-1)*(sqrt(1-chebyshev_T(2*j,cs)^2))/(u^2+1)
    if L_approx(wx)<0: wx=-wx
    show(wx) #Looking at the expressions as polynomials in u or u and v
    show(wy)
```

```
print L_approx(wx), L_approx(wy) #Numerical values
```

```
35/64*u^11 - 91/64*u^9 + 7/32*u^7 - 83/32*u^5 + 31/64*u^3 - 15/64*u
-21/64*u^10 + 35/64*u^8 + 35/32*u^6 + 19/32*u^4 + 15/64*u^2 - 41/64
0.215627386401338 0.447754881549269
```

---

12) We define the function  $g(z) = \frac{z-w}{\bar{w}z-1} = \frac{(z-w)(w\bar{z}-1)}{|w|^2|z|^2-2re(w\bar{z})+1} = \frac{x+iy-wx-iwy}{(wx-iwy)(x+iy)-1}$ , but rather as a function  $\mathbb{R}^2 \rightarrow \mathbb{R}^2$ . This is for technical reasons (L has no chosen embedding into the complex numbers, so we couldn't perform in principle products as  $u \cdot i$  that we will later need).

We need to compute the point  $p = \frac{g(zx+izy)}{1+\sqrt{1-|zx+izy|^2}}$ , but the denominator is never in L because  $1-|zx+izy|^2$  is never a square. So we extend the field L to  $K2 = L(w)$ , where  $w = \sqrt{1-|zx+izy|^2}$ .

Finally  $m = g(p)$  may be, according to our numerical experiments, the fixed point of an order two elliptic element inducing an automorphism of the surface.

```
[36]: def g(x,y):
    denom=(x*wx+y*wy-1)^2+(y*wx-x*wy)^2
    num1=(x^2+y^2)*wx-x*(wx^2-wy^2)-2*y*wx*wy-x+wx
    num2=(x^2+y^2)*wy-2*x*wx*wy+y*(wx^2-wy^2)-y+wy
    return [num1/denom, num2/denom]

show((1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2)) #Inspecting how does w^2 look like

if (1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2).is_square():
    print 'Warning: apparently w IS in the field L' #Apparently this never
    →happens in our files
    #If it happens the last lines of this cell would produce an error, since
    →K2_approx
    #wouldn't be defined
else:
    Ring.<y> = PolynomialRing(L)
    K2.<omega> = L.extension(y^2-(1-(g(zx,zy)[0])^2-(g(zx,zy)[1])^2))
    px=g(zx,zy)[0]/(1+omega)
    py=g(zx,zy)[1]/(1+omega)
    mx=g(px,py)[0]
    my=g(px,py)[1]
    show(mx) #These are the coordinates of the fixed point of tau as elements of
    →the field L.
    show(my)
    #Again choosing a precise embedding of K2 into the complex numbers
```

```

#This is not really needed, but allows the computation of numerical values
for mx, my
j=0
test=0
while test!=1:
    K2_approx = K2.embeddings(CC) [j]
    if abs(CC(sqrt(u^4+u^2+1)/(u^2+1))-K2_approx(v))<10^-3 and
abs(CC(u)-K2_approx(u))<10^(-3) and K2_approx(omega)>0: test+=1
    j+=1
    print K2_approx(u),K2_approx(v), K2_approx(omega) #This line must give the
values we knew in advance (numerical verification)

print K2_approx(mx), K2_approx(my)

```

```

49/4*u^10 - 133/4*u^8 + 21/4*u^6 - 91/2*u^4 + 3*u^2 + 21/4
(-35/688*u^11 + 147/688*u^9 - 203/688*u^7 + 187/688*u^5 - 2/43*u^3 + 20/
 $\downarrow$ 43*u)*omega - 7/5504*u^11 - 1897/5504*u^9 + 2779/2752*u^7 + 109/2752*u^5 + 2737/
 $\downarrow$ 5504*u^3 - 4881/5504*u

(-511/2752*u^10 + 189/688*u^8 + 245/688*u^6 + 1377/1376*u^4 + 1963/2752*u^2 + 725/
 $\downarrow$ 1376)*omega + 371/2752*u^10 - 553/2752*u^8 - 63/1376*u^6 - 1577/1376*u^4 - 1973/
 $\downarrow$ 2752*u^2 - 2533/2752

1.72508268711722 0.900968849766670 0.973703548079917
0.128823136760549 0.422223869592106

```

---

### 13) We compute the coordinates of $\bar{\tau}$

[37] :

```

tau00=(-2*mx)/(1-mx^2-my^2)
tau01=(2*my+1+mx^2+my^2)/(1-mx^2-my^2)
tau10=(2*my-1-mx^2-my^2)/(1-mx^2-my^2)
tau11=(2*mx)/(1-mx^2-my^2)
tau=matrix(2,[tau00,tau01,tau10,tau11])
show(tau) # Once more we are curious. The entries lie in K2, but: how do they
 $\downarrow$ look like?

```

```

[(-775/5312*u^11 + 927/5312*u^9 + 1105/2656*u^7 + 15469/18592*u^5 + 2789/5312*u^3
 $\downarrow$ + 1483/5312*u)*omega (-731/5312*u^10 + 1707/5312*u^8 + 383/2656*u^6 + 9875/
 $\downarrow$ 18592*u^4 + 1277/5312*u^2 + 1579/5312)*omega]
[ (3045/5312*u^10 - 5733/5312*u^8 - 1617/2656*u^6 - 8411/2656*u^4 - 6627/
 $\downarrow$ 5312*u^2 - 9509/5312)*omega (775/5312*u^11 - 927/5312*u^9 - 1105/2656*u^7 - 15469/18592*u^5 - 2789/5312*u^3 - 1483/5312*u)*omega]

```

---

14) We create a list  $\text{conj}$  containing what our numerical experiments suggested that should be  $c_j$ , the conjugate by  $\tau$  of the side-pairing  $s_j$ , for each  $j$ . We finally check that  $\tau s_j \tau^{-1} c_j^{-1}$  is trivial.

```
[38]: conj=['Zero'] #Create this list with the data from the numerical experiments!!!!
conj.append(s[9]^(-1)*s[1]^(-1)*s[9])
conj.append(s[6]*s[9])
conj.append(s[6]*s[9])
conj.append(s[6]*s[9])
conj.append(s[6]^(-1))
conj.append(s[6]^(-1))
conj.append(s[6]^(-1))
conj.append(s[9]^(-1))
conj.append(s[9]^(-1))
conj.append(s[9]^(-1)*s[7])
conj.append(s[9]^(-1)*s[11]*s[7])
conj.append(s[9]^(-1)*s[11]*s[7])

for j in [1..len(s)-1]:
    show(tau*s[j]*tau^(-1)*conj[j]^(-1))
```

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$

$[1 \ 0]$   
 $[0 \ 1]$

[ ]: